2.　項目分析における項目統計量と構成尺度の統計量

——因子的真実性係数と因子的妥当性——

〔問　題〕

われわれは先の共同研究「確認的因子分析における検査尺度構成」において（辻岡,1975），質問紙法形式の性格検査や態度検査，興味検査などにおける項目分析においては，項目間相関の因子分析による項目選択ではなく，比較的等質性の高い習性水準尺度バッテリーの因子分析によって得られた一次因子との間において，因子的真実性の高い項目群からなる尺度を構成する方法について説明した。本論文ではこの構想を完成させ，最終的に選択された項目群尺度の因子的妥当性係数 (coefficient of factorial validity) や因子的真実性係数 (coefficient of factor-trueness)（辻岡，1964 および Cattell and Tsujioka, 1964）を算出して，構成尺度の因子的一義性の成否を評価する方法を展開したいと考える。なお，項目選択のためのプログラム（辻岡・清水，1975）はすでに発表したが，ここでは本論の理論展開にそって改良したプログラム，すなわち，欲する因子と項目変量との斜交因子パターン行列を求めるためのメインプログラム，項目選択のサブルーチンと因子的真実性係数や因子的妥当性係数のためのサブルーチンをも掲載し，研究者の便をはかりたいと考える。

先の論文においても説明したように，三件法や二件法の項目変量では triserial correlalation（三系列相関）や biserial correlation（二系列相関）が必要となる。しかし，ここでは，少なくとも五件法以上の項目のための単なる積率相関係数を用いた方法でプログラムが書かれている。もし，三件法以下の場合には，先の論文（辻岡・藤村，1975）で発表した multiserial correlation のサブルーチンを結合して使用されることをおすすめする。

なお，ここで開発したプログラムは，第二章第 4 節の村山による価値観尺度の構成のために応用された。したがって，掲載された研究例については村山の節を参照しながら読んで戴ければ具体的な理解に役立つと考える。

〔方　法〕

(1)　項目分析のための項目統計量

われわれの因子分析は，項目間相関行列を直接因子分析するのではなく，数項目からなる習性水準の尺度間相関行列に対して行なわれる。そこで，項目分析のためには，習性水準の尺度間相関の因子分析によって求められた因子軸空間にこれらの項目変量を投影する必要がある。すなわち，各項目変量がどのような因子的組成からなるかを調べるわけであり，これは別名延長因子分析 (extension factor analysis) ともよばれるものである。

そこで，項目変量を因子軸空間に投影するために，出発尺度間の因子分析より因子得点を求める。しかしながら，実際に算出されるのは真の因子得点ではなく因子推定値行列 $\bar{\mathbf{F}}$ であり，その計算は次の式によって行なわれる（なお，この式は芝（1972）の F_{13} 式に該当する）。

(1) $\overline{\mathbf{F}}=\mathbf{Z}_s\mathbf{R}_s^{-1}\mathbf{V}_{fs}$

ここで，$\overline{\mathbf{F}}$ は（N×m）次の因子推定値行列，$\mathbf{Z}_s$ は（N×n）次の出発尺度の標準得点行列，$\mathbf{R}_s^{-1}$ は（n×n）次の出発尺度間相関行列の逆行列，$\mathbf{V}_{fs}$ は（n×m）次の因子構造行列である。ただし，Nは被験者数，nは尺度数，mは因子数である。

ここで求まるものは推定値であって真の因子得点ではない。この因子推定値（$\overline{\mathrm{F}}$）と真の因子得点（F）との関係は，

(2) $r_{F\overline{F}}=\sigma_{\overline{F}}$

により求められる。すなわち，真の因子得点（F）と推定値（$\overline{\mathrm{F}}$）との相関は，$\overline{\mathrm{F}}$の標準偏差である。この値が0.8以上あれば，この推定は成功的と言える。

さて，延長因子分析によって求まる項目の因子構造 $\hat{\mathbf{V}}_{fs}$ は，項目変量と因子推定値との相関であるから，

(3) $\hat{\mathbf{V}}_{fs}=\frac{1}{N}\mathbf{Z}_i'\overline{\mathbf{F}}\mathbf{S}^{-1}$

で算出される。項目の因子パターン $\hat{\mathbf{V}}_{fp}$ は，

(4) $\hat{\mathbf{V}}_{fp}=\hat{\mathbf{V}}_{fs}\mathbf{C}_f^{-1}$

によって求まる。ここで，因子推定値の平均は０であり，$\mathbf{S}$は因子推定値の標準偏差 $\sigma_{\overline{F}}$ を対角に含む（m×m）次の対角行列とする。また，$\hat{\mathbf{V}}_{fs}$ は（l×m）次の推定因子構造行列，$\hat{\mathbf{V}}_{fp}$ は（l×m）次の推定因子パターン行列，$\mathbf{Z}_i$ は（N×l）次 の項目の標準得点行列，$\mathbf{C}_f$ は（m×m）次の習性尺度間因子分析によって求まっている一次因子間相関行列である。なお，lは選択項目数である。

また，項目の因子構造を求める時，上式(3)とは別な計算手順も考えられる。すなわち，

(5) $\frac{1}{N}\mathbf{Z}_i'\overline{\mathbf{F}}\mathbf{S}^{-1}=\frac{1}{N}\mathbf{Z}_i'\mathbf{Z}_s\mathbf{R}_s^{-1}\mathbf{V}_{fs}\mathbf{S}^{-1}$

ここで，因子推定値を求めるための標準重み行列は，

(6) $\hat{\mathbf{W}}=\mathbf{R}_s^{-1}\mathbf{V}_{fs}\mathbf{S}^{-1}$

であり，

(7) $\hat{\mathbf{F}}=\mathbf{Z}_s\hat{\mathbf{W}}$

となる。ここで，$\hat{\mathbf{F}}$ を標準因子推定値行列（N×m）とよぶ。今，

(8) $\mathbf{R}_{is}=\frac{1}{N}\mathbf{Z}_i'\mathbf{Z}_s$

とすると，式(3)と(5)より，

(9) $\hat{\mathbf{V}}_{fs}=\mathbf{R}_{is}\hat{\mathbf{W}}$

となる。ここで，$\mathbf{R}_{is}$ は（l×n）次の項目・出発尺度間相関行列であり，$\hat{\mathbf{W}}$ は（n×m）次の因子得点のための標準重み行列である。当然，推定因子パターン行列は式(4)によって求まる（$\mathbf{W}$と$\hat{\mathbf{W}}$ の違いに注意）。

(2) 構成尺度の統計量

（ⅰ）構成尺度の平均と分散

今，二件法，三件法あるいは五件法等の項目変量を個人 p（p =1, 2, ……, N）については行に，項目 j（j =1, 2, ……, l）については列にならべた行列を $\mathbf{X}$（N× l）とする。そのとき個人 p の構成尺度得点 Y_p は，

(10) $$Y_p = \sum_{j=1}^{l} X_{pj}$$

であり，N人分をベクトルであらわすと，

(11) $$\mathbf{y} = \mathbf{X1}_l$$

ここで，$\mathbf{1}_l$ は l 次の 1 からなる列ベクトルである。したがって，この構成尺度のN人分の平均は，

(12) $$\bar{y} = \frac{1}{N}\sum_{p=1}^{N} Y_p = \frac{1}{N}\sum_{p=1}^{N}\sum_{j=1}^{l} X_{pj}$$

となり，これは行列表記では

Table 2－1 得点行列

		項目1	項目2	……	項目 j	……	項目 l	構成尺度粗点
個人	1	X_{11}	X_{12}	……	X_{1j}	……	X_{1l}	$\sum_{j=1}^{l} X_{1j} = Y_1$
	2	X_{21}	X_{22}	……	X_{2j}	……	X_{2l}	$\sum_{j=1}^{l} X_{2j} = Y_2$
	⋮	⋮	⋮		⋮		⋮	⋮
	P	X_{p1}	X_{p2}	……	X_{pj}	……	X_{pl}	$\sum_{j=1}^{l} X_{pj} = Y_p$
	⋮	⋮	⋮		⋮		⋮	⋮
	N	X_{N1}	X_{N2}	……	X_{Nj}	……	X_{Nl}	$\sum_{j=1}^{l} X_{Nj} = Y_N$
項目平均		$\frac{\sum_{p=1}^{N} X_{p1}}{N}$	$\frac{\sum_{p=1}^{N} X_{p2}}{N}$	……	$\frac{\sum_{p=1}^{N} X_{pj}}{N}$	……	$\frac{\sum_{p=1}^{N} X_{pl}}{N}$	$\frac{1}{N}\sum_{p=1}^{N}\sum_{j=1}^{l} X_{pj} = \frac{1}{N}\sum_{p=1}^{N} Y_p$（尺度平均）

(13) $$\bar{y} = \frac{1}{N}\mathbf{1}_N'\mathbf{y} = \frac{1}{N}\mathbf{1}_N'\mathbf{X1}_l$$

となる。ここで，$\mathbf{1}_N'$ はN個の 1 からなる行ベクトルである。

次に，構成尺度の分散は，

(14) $$\sigma_y{}^2 = \frac{1}{N}\sum_{p=1}^{N}(y_p - \bar{y})^2$$

であり，行列表記では，

(15) $$\sigma_y{}^2=\frac{1}{N}(\mathbf{y}-\mathbf{1}_N\bar{y})'(y-\mathbf{1}_N\bar{y})$$
$$=\frac{1}{N}\left(\mathbf{X1}_l-\mathbf{1}_N\frac{1}{N}\mathbf{1}_N'\mathbf{X1}_l\right)'\left(\mathbf{X1}_l-\mathbf{1}_N\frac{1}{N}\mathbf{1}_N'\mathbf{X1}_l\right)$$
$$=\frac{1}{N}\left(\mathbf{1}_l'\mathbf{X}'-\mathbf{1}_l'\frac{1}{N}\mathbf{X}'\mathbf{1}_N\mathbf{1}_N'\right)\left(\mathbf{X1}_l-\mathbf{1}_N\frac{1}{N}\mathbf{1}_N'\mathbf{X1}_l\right)$$
$$=\frac{1}{N}\mathbf{1}_l'\left(\mathbf{X}'-\frac{1}{N}\mathbf{X}'\mathbf{1}_N\mathbf{1}_N'\right)\left(\mathbf{X}-\mathbf{1}_N\frac{1}{N}\mathbf{1}_N'\mathbf{X}\right)\mathbf{1}_l$$
$$=\frac{1}{N}\mathbf{1}_l'\boldsymbol{\Sigma}_i\boldsymbol{\Sigma}_i{}^{-1}\left(\mathbf{X}'-\frac{1}{N}\mathbf{X}'\mathbf{1}_N\mathbf{1}_N'\right)\left(\mathbf{X}-\mathbf{1}_N\frac{1}{N}\mathbf{1}_N'\mathbf{X}\right)\boldsymbol{\Sigma}_i{}^{-1}\boldsymbol{\Sigma}_i\mathbf{1}_l$$

となる。ただし，ここで $\boldsymbol{\Sigma}_i$ は（$l\times l$）次の項目変量の標準偏差を主対角にもつ対角行列である。したがって，

(16) $$\boldsymbol{\Sigma}_i{}^{-1}\frac{1}{N}\left(\mathbf{X}-\mathbf{1}_N\frac{1}{N}\mathbf{1}_N'\mathbf{X}\right)'\left(\mathbf{X}-\mathbf{1}_N\frac{1}{N}\mathbf{1}_N'\mathbf{X}\right)\boldsymbol{\Sigma}_i{}^{-1}=\mathbf{R}_i$$ となり，

(17) $$\sigma_y{}^2=\mathbf{1}_l'\boldsymbol{\Sigma}_i\mathbf{R}_i\boldsymbol{\Sigma}_i\mathbf{1}_l$$

となる。しかるに，

(18) $$\mathbf{1}_l'\boldsymbol{\Sigma}_i=\boldsymbol{\sigma}_i'$$

となる。ここで，$\boldsymbol{\sigma}_i$ は項目変量の標準偏差ベクトルである。故に，

(19) $$\sigma_y{}^2=\boldsymbol{\sigma}_i'\mathbf{R}_i\boldsymbol{\sigma}_i$$

となる。すなわち，構成尺度（Y）の分散は，項目変量間相関行列の前後に項目変量の標準偏差ベクトルを乗じたものとなる。ここで出発尺度（S）とこの項目分析の結果作成される構成尺度（Y）の記号とを混合してはならない。また，ｉの添字は項目のためのものである。

(ii) 構成尺度の因子構造（因子的妥当性係数）



Fig. 2－1
構成尺度の共通性，因子構造，因子パターン

今，構成尺度と一次因子との相関は，所謂因子構造（factor structure）であるが，これはまた別名因子的妥当性係数(coefficient of factorial validity）ともよばれる。一方，構成尺度の斜交因子軸に対する座標は，因子パターン（factor pattern）とよばれる。

今，この両者の関係は Fig. 2－1の平面によって説明できる。ここで，ベクトルOTは，共通因子推定値空間 $\bar{F}_p\cdot\bar{F}_q$ 面における構成尺度ベクトルをあらわし，OS は因子推定値軸 $O\bar{F}_q$ へのこのOTの射影である。すなわち，

(20) $$OS=r_{qy}$$

である。OSはこの構成ベクトルと因子推定値軸 $\bar{F}_q$ との相関を意味する。

そこで，この値を項目統計量から求めるため，

(21) $$\overline{\mathbf{f}}_q \mathbf{S}_q^{-1} = \hat{\mathbf{f}}_q$$

として因子推定値を標準化し，これをN人分のベクトルとすると，

(22) $$r_{qy} = \frac{1}{N} \hat{\mathbf{f}}_q'(\mathbf{y} - \mathbf{1}_N \overline{y})\sigma_y^{-1}$$

となる。ここで，$(\mathbf{y} - \mathbf{1}_N \overline{y})$ はN人分の構成尺度の平均 $\overline{y}$ よりの偏差であり，σ_y^{-1} は構成尺度の標準偏差の逆数（スカラー）である。上の(22)式は，

(23) $$r_{qy} = \frac{1}{N\sqrt{\boldsymbol{\sigma}_i' \mathbf{R}_i \boldsymbol{\sigma}_i}}(\hat{\mathbf{f}}_q'\mathbf{y} - \hat{\mathbf{f}}_q'\mathbf{1}_N \overline{y})$$

となり，上式のカッコの中の第二項中，

(24) $$\hat{\mathbf{f}}_q'\mathbf{1}_N = 0$$

であり，(11)式を代入して整理すると

(25) $$r_{qy} = \frac{\hat{\mathbf{f}}_q'\mathbf{y}}{N\sqrt{\boldsymbol{\sigma}_i' \mathbf{R}_i \boldsymbol{\sigma}_i}} = \frac{\hat{\mathbf{f}}_q'\mathbf{X}\mathbf{1}_l}{N\sqrt{\boldsymbol{\sigma}_i' \mathbf{R}_i \boldsymbol{\sigma}_i}}$$

となる。ところが，

(26) $$\mathbf{X} = (\mathbf{X} - \mathbf{1}_N \mathbf{1}_N' \mathbf{X}/N + \mathbf{1}_N \mathbf{1}_N' \mathbf{X}/N)$$

を(25)式に代入すると分子は，

(27) $$\hat{\mathbf{f}}_q'(\mathbf{X} - \mathbf{1}_N \mathbf{1}_N' \mathbf{X}/N)\mathbf{1}_l + \hat{\mathbf{f}}_q'\mathbf{1}_N \mathbf{1}_N' \mathbf{X}/N\mathbf{1}_l = \hat{\mathbf{f}}_q'(\mathbf{X} - \mathbf{1}_N \mathbf{1}_N' \mathbf{X}/N)\mathbf{1}_l$$

となる。なんとなれば，上の左辺第二項は再び

(28) $$\hat{\mathbf{f}}_q'\mathbf{1}_N = 0$$

となり，消失するからである。上式(27)の右辺は，

(29) $$\hat{\mathbf{f}}_q'(\mathbf{X} - \mathbf{1}_N \mathbf{1}_N' \mathbf{X}/N)\boldsymbol{\Sigma}_i^{-1}\boldsymbol{\Sigma}_i \mathbf{1}_l$$

と書ける。ここで，$\boldsymbol{\Sigma}_i$ は項目変量の標準偏差を主対角にもつ対角行列である。したがって，

(30) $$r_{qy} = \frac{\hat{\mathbf{f}}_q'(\mathbf{X} - \mathbf{1}_N \mathbf{1}_N' \mathbf{X}/N)\boldsymbol{\Sigma}_i^{-1}}{N} \cdot \frac{\boldsymbol{\Sigma}_i \mathbf{1}_l}{\sqrt{\boldsymbol{\sigma}_i' \mathbf{R}_i \boldsymbol{\sigma}_i}}$$

$$= \frac{1}{N}\hat{\mathbf{f}}_q'\mathbf{Z}_i \cdot \frac{\boldsymbol{\Sigma}_i \mathbf{1}_l}{\sqrt{\boldsymbol{\sigma}_i' \mathbf{R}_i \boldsymbol{\sigma}_i}}$$

$$= \mathbf{r}_{qi}' \cdot \frac{\boldsymbol{\sigma}_i}{\sqrt{\boldsymbol{\sigma}_i' \mathbf{R}_i \boldsymbol{\sigma}_i}}$$

となる。ここで，$\mathbf{r}_{qi}$ は因子推定値 $\overline{F}_q$ と項目との相関ベクトルであり，これは項目と因子推定値との矩形相関行列 $\hat{\mathbf{V}}_{fs}$（項目×因子）のうちのq番目についての列ベクトルの一部 $\hat{\mathbf{v}}_{fs}^*$ の転置である。すなわち，

(31) $$(\mathbf{r}_{qi}')' = \hat{\mathbf{v}}_{fs}^*$$ とすると，

(32) $$r_{qy} = \frac{\boldsymbol{\sigma}_i'\hat{\mathbf{v}}_{fs}^*}{\sqrt{\boldsymbol{\sigma}_i' \mathbf{R}_i \boldsymbol{\sigma}_i}}$$

となる。この式の分子はスカラーであるから上式の分子を転置しても等しいと考えてもよい。

(iii) 構成尺度の因子パターン

今，項目変量の推定因子構造行列を $\hat{\mathbf{V}}_{fs}$ とし，推定因子パターン行列を $\hat{\mathbf{V}}_{fp}$ とすると，

(33) $$\hat{\mathbf{V}}_{fp}=\hat{\mathbf{V}}_{fs}\mathbf{C}_f{}^{-1}$$

であることは，先にものべた。上の $\hat{\mathbf{v}}_{fs}{}^*$ は $\hat{\mathbf{V}}_{fs}$ の中から選択項目のみの部分を列ベクトルの形でとり出したものである。

今，選択項目とすべての因子推定値との相関すなわち選択推定因子構造行列を $\hat{\mathbf{V}}_{fs}{}^*$ とし，それに対応する因子パターン行列を $\hat{\mathbf{V}}_{fp}{}^*$ とすると，

(34) $$\hat{\mathbf{V}}_{fp}{}^*=\hat{\mathbf{V}}_{fp}{}^*\mathbf{C}_f{}^{-1}$$

となる。故に，構成尺度（S）の推定因子パターンベクトル ${}_s\hat{\mathbf{v}}_{fp}{}^*$ は，

(35) $${}_s\hat{\mathbf{v}}_{fp}{}^{*\prime}=\frac{\boldsymbol{\sigma}_i{}'\hat{\mathbf{V}}_{fs}{}^*}{\sqrt{\boldsymbol{\sigma}_i{}'\mathbf{R}_i\boldsymbol{\sigma}_i}}\mathbf{C}_f{}^{-1}=\frac{\boldsymbol{\sigma}_i{}'(\hat{\mathbf{V}}_{fs}{}^*\mathbf{C}_f{}^{-1})}{\sqrt{\boldsymbol{\sigma}_i{}'\mathbf{R}_i\boldsymbol{\sigma}_i}}$$
$$=\frac{\boldsymbol{\sigma}_i{}'\hat{\mathbf{V}}_{fp}{}^*}{\sqrt{\boldsymbol{\sigma}_i{}'\mathbf{R}_i\boldsymbol{\sigma}_i}}$$

となり，これがＯＰの長さにひとしくなる。換言すれば，構成尺度の因子パターンは項目の標準偏差によって重みづけられた項目の因子パターンを構成尺度の標準偏差によって標準化したものとなる。

(iv) 構成尺度の共通性と因子的真実性係数

次に，Fig. 2－1の（OT)^2 すなわち，構成尺度の共通性は次の考え方を適用すればすぐに求められる。

すなわち，共通性 h^2 は一般に，

(36) $$h^2=\mathbf{v}_{fs}\mathbf{v}_{fp}{}'$$

である。ここで，$\mathbf{v}_{fs}$ および $\mathbf{v}_{fp}$ は l 次のある測定変量の因子構造ベクトルと因子パターンベクトルである。

したがって，構成尺度の共通性 $h_q{}^2$ は，

(37) $$h_q{}^2=\frac{\boldsymbol{\sigma}_i{}'\hat{\mathbf{v}}_{fs}{}^*}{\sqrt{\boldsymbol{\sigma}_i{}'\mathbf{R}_i\boldsymbol{\sigma}_i}}\cdot\left(\frac{\boldsymbol{\sigma}_i{}'\hat{\mathbf{v}}_{fp}{}^*}{\sqrt{\boldsymbol{\sigma}_i{}'\mathbf{R}_i\boldsymbol{\sigma}_i}}\right)'=\frac{\boldsymbol{\sigma}_i{}'\hat{\mathbf{v}}_{fs}{}^*\hat{\mathbf{v}}_{fp}{}^{*\prime}\boldsymbol{\sigma}_i}{\boldsymbol{\sigma}_i{}'\mathbf{R}_i\boldsymbol{\sigma}_i}$$

となる。したがって，ベクトルのノルムＯＴは，

(38) $$OT=\frac{\sqrt{\boldsymbol{\sigma}_i{}'\hat{\mathbf{v}}_{fs}{}^*\hat{\mathbf{v}}_{fp}{}^{*\prime}\boldsymbol{\sigma}_i}}{\sqrt{\boldsymbol{\sigma}_i{}'\mathbf{R}_i\boldsymbol{\sigma}_i}}$$ となる。

故に，因子的真実性係数 (coefficient of factor-trueness) r_{ft} は，

(39) $$r_{ft}=\cos\theta=\frac{OS}{OT}$$
$$=\frac{\boldsymbol{\sigma}_i{}'\hat{\mathbf{v}}_{fs}{}^*}{\sqrt{\boldsymbol{\sigma}_i{}'\mathbf{R}_i\boldsymbol{\sigma}_i}}\Bigg/\frac{\sqrt{\boldsymbol{\sigma}_i{}'\hat{\mathbf{v}}_{fs}{}^*\hat{\mathbf{v}}_{fp}{}^{*\prime}\boldsymbol{\sigma}_i}}{\sqrt{\boldsymbol{\sigma}_i{}'\mathbf{R}_i\boldsymbol{\sigma}_i}}$$

$$=\frac{\boldsymbol{\sigma}_i'\hat{\mathbf{V}}_{fs}^*}{\sqrt{\boldsymbol{\sigma}_i'\hat{\mathbf{V}}_{fs}^*\hat{\mathbf{V}}_{fp}^{*\prime}\boldsymbol{\sigma}_i}}$$

によって計算することができる。

〔考　察〕

項目分析は，項目の選択が完了すれば当然構成尺度が出来上る。上の種々の公式は構成尺度が出来上ってしまえば，わざわざ項目統計量にもどって計算しなくても，構成尺度の統計量からこれらをもっと簡単に計算することができる。しかし，ここで将来出来上る構成尺度の平均，分散，因子構造，因子パターン，因子的真実性などを項目統計量から導出したのは，将来出来上るべき尺度（全体）と既知の項目（部分）統計量との関係を，われわれが前以って明示的に認識しておく必要があったからである。

ここで求められた因子的真実性と因子的妥当性（因子構造）とは，一方が＋１に近づけば近づくほど他も＋１になる。しかし，いずれもが0.7～0.8の水準において，項目の選択に関していずれか一方を 高くしようとすれば， 他はそれだけ 犠牲にしなければならぬということも 起ってくる。たとえば，因子構造値は高いが因子真実性の低い項目を選ぶか，それとも多少因子構造値は低いが因子的真実性の高い項目を選ぶかのジレンマがおこってくる。このいずれを選ぶかについては，未だ客観的な基準が定まっていない。今後はこの点についての検討を行いたいと考えている。この点抑制の原理をうまく働かしうる，一定範囲内のすべての項目の組合せを考えて尺度構成を行うプログラムも考えられる。たとえば，今ｎ個の項目のプールがあるとき，l 個を選択するとすれば ${}_n\mathbf{C}_l$ 通りがあり，このすべてのうちから最善の一通りを選択する方法が考えられる。Fig. 2－2は，因子的真実性係数と因子的妥当性係数とをプロットした図である。そして，この中から最善のものを選ぶ賢明な客観的基準を考案したいと考えている。



Fig. 2－2
構成尺度の因子的妥当性と因子的真実性

〔プログラムの説明〕

(1) **MAIN PROGRAM** について

このプログラムは推定因子構造 $\hat{\mathbf{V}}_{fs}$ と推定因子パターン $\hat{\mathbf{V}}_{fp}$ とを求めるものである。Fig. 2－3の通り，ここでは(3)式によてっ $\hat{\mathbf{V}}_{fs}$ を求めた。その際，磁気テープに粗点と因子推定値とが各被験者について組み合せて書き込まれているものを用いている。このようなテープの利用が不可能な場合には，(8)式と(9)式を用いる方が便利であろう。というのは，(8)式の $\mathbf{Z}_s$ は $\mathbf{Z}_i$ の中から構成される尺度であるからである。また，(3)式によるか(8)式によるかは，電算機の容量と

INPUT DATA の説明

NX	項目の数（Ｉ４）
NF	因子の数（Ｉ４）
EN	被験者の終りを示す数（F8.0）
ITTL	各ページに研究題名を印字するもの（40Ａ２）
NMF	各因子の名前を２文字で表わしたもの(20(2X，A2))
NFI	各項目の名前を２文字で表わしたもの(20(2X，A2))
S	各因子推定値の標準偏差（10Ｆ8.0）
X	項目得点（テープより）
F	因子推定値（テープより）
CF	一次因子間相関行列（10Ｆ8.0）（左下三角行列）
CR	準拠軸間相関行列（10Ｆ8.0）（左下三角行列）

OUTPUT DATA の説明

AVF	標準因子推定値の平均
SDF	標準因子推定値の標準偏差
CF	標準因子推定値間の相関行列
AVX	項目の平均
SDX	項目の標準偏差
EVFS	推定因子構造行列
EVFP	推定因子パターン行列



Fig. 2－3
メインプログラムの流れ図

もかかわる。前式は（$l \times m$）次の計算であり，後式は（$l \times n$）次の計算となる。ここで，mはnの約⅓程度であり，その節約は相当大きなものである。

なお，逆行列を求めるための **SUBROUTINE MATINV** は浅野(1971) を引用した。**SUBROUTINE PLOTK** は，$\hat{\mathbf{V}}_{fp}$ を二次元平面において ${}_mC_2$ 枚印字するものであり辻岡・藤村 (1975) よりの引用である。準拠軸間相関行列と因子間相関行列は，このプロットにおいて必要とされるので読み込ませてある。また，**SUBROUTINE MOUT** は行列を印字させるためのもであり，利用者が用意されることを望む。

(2) **SUBROUTINE ISFT** について

これは，推定因子パターン行列 $\hat{\mathbf{V}}_{fp}$ を用いて項目選択を行なうサブルーチンである。

INPUT DATA 及び引数の説明

NX	項目数
NF	因子数
FVFP	推定因子パターン行列
EVFS	推定因子構造行列
SD	項目の標準偏差
IFIN	項目選択を自動で行なう（0を入れた場合）
	項目番号を指定する（0以外の値の場合）
JJ	選択された項目数（I 4）
NN	選択された項目番号，逆転項目にはマイナス符号を付ける（20 I 4）

OUTPUT DATA の説明

EVEP	選択された推定因子パターン $\hat{\mathbf{V}}_{fp}{}^*$
NMI	その項目名
J	その項目番号
IH	逆項目に ＊ を印字
IZ	当該因子以外の因子において選択されていたら，その因子番号を印字
AVE	各因子において選択された $\hat{\mathbf{V}}_{fp}{}^*$ の平均値
RFT	因子的真実性係数

項目選択の手続は，尺度構成の対象となる因子に対する因子パターンの絶対値がＣＯＮ以上あり，かつこの値をｗとしその他の因子に対する因子パターンの絶対値をｕとすると $\tan\theta=\dfrac{u}{w}$ という関係から当該因子軸の $\theta°$（TN）内に位置する項目であること，またこの角度の範囲内にすべての他の因子との組み合せにおいてNM回以上入る，という基準の下で行なわれる。



Fig. 2－4 SUBROUTINE ISFT の流れ図

なお，これらの CON, TN, NM は，メインプログラムにおいて DATA 文で与えてあるが，このCONを0.7という高い値としたのは，それよりも小さくすると選択される項目数が多くなり，このサブルーチンにおける NN, NH, IH, や次のサブルーチンにおけるR等の DIMENSION が大きくなりすぎ，われわれの使用する FACOM 230―25/35 のメモリー容量を越えるからである。

(3) **SUBROUTINE SPC** について

これは，構成尺度分散を算出し，(32)式，(35)式，(37)式そして(39)式を実行して構成尺度の因子的妥当性係数（因子構造），因子パターン，共通性と因子的真実性係数を求めるためのものである。

ここで必要な項目の平均と標準偏差はすでにメインプログラムにおいて求まっているので引数で対応させてそれを用いており，このサブルーチンでは項目間の積和のみを計算させている。**ISFT** において逆転された項目については，その平均値は **ISFT** 内で逆転してあるので，ここでは項目得点の逆転を行なっている。

各因子ごと (**ISFT** におけるK）に粗点を読み直す方式を取った。ここで，Fとして因子推定値を読み込ませてはいるが，これはから読みであって計算とは関係ない。テープが利用出来ない場合には，すべての因子についての項目選択が終ってから，項目間相関を算出

INPUT DATA 及び引数の説明

NX	項目の数
NF	因子の数
EN	被験者の終りを示す数
JJ	選択された項目数
EVFP	推定因子パターン行列
EVFS	推定因子構造行列
AV	項目の平均（逆転項目は **ISFT** 内で逆転ずみ）
SD	項目の標準偏差
NN	**ISFT** で選択された項目番号（逆転項目はマイナス符号）
NH	**ISFT** で選択された項目番号
K	尺度構成の対象となっている因子番号
X	項目の粗点（テープより）



Fig. 2－5
SUBROUTINE SPCの流れ図

OUTPUT DATA の説明

OS	因子的妥当性係数（因子構造）
OP	因子パターン
OT 2	共通性（h^2）
OT	構成尺度のベクトルの長さ（h）
RFT	因子的真実性係数
R	項目間相関行列

する方法を考える必要がある。

なお，このサブルーチンは，項目分析の最終段階で用いるものであろう。というのは，前にも述べたように，ここではRという項目間相関のための領域が必要であり，分析の出発点ではこの大きさを想定することは困難である。そこで，**ISFT** において，NN，NH，IN を大きめにとり，ＣＯＮをやや低くめにして，この **SPC** を呼ばずに，まず，因子的真実性係数や **PLOTK** における布置図を手掛りに全体的な見通しを把握し，選択の基準を変えながら項目分析を繰り返し，項目の内容から因子解釈の精密化をはかることが望ましい。

〔要　約〕

1. 因子的真実性の原理による項目分析のための電算プログラムの数学的基礎が展開された。

2. 項目統計量と，項目選択によって構成された構成尺度の統計量，すなわち構成尺度の平均，分散，因子構造，因子パターン，共通性，因子的真実性係数との数学的関係が明示された。

3. このための電算プログラムが FORTRAN 言語によって開発された。それらはメインプログラムと **ISFT**，**SPC** のサブプログラムから成る。

4. 因子的真実性係数と因子的妥当性の高さの競合的矛盾を解決し，最大の因子的真実性と因子的妥当性をうるための解決策が考察された。

〔参　考　文　献〕

1. 浅野長一郎　1971　因子分析法通論　共立出版.
2. Cattell, R. B., and Tsujioka, B. 1964 The importance of factor-trueness and validity versus homogeneity and orthogonality, in test scales. *Educ. Psychol. Measurement,* **24**, 3—30.
3. 芝　祐順　1972　因子分析法　東京大学出版会.
4. 辻岡美延　1964　テスト尺度構成における新しい原理—因子的真実性—心理学評論，**8**，82—90.
5. 辻岡美延　1975　確認的因子分析における検査尺度構成（問題と方法—習性水準尺度を出発尺度とする検査尺度構成について—）関西大学社会学部紀要，**6**(1)，5—14.
6. 辻岡美延・藤村和久　1975　確認的因子分析における検査尺度構成（項目分析のための相関係数—多分相関および多系列相関について—）関西大学社会学部紀要，**6**(1)，15—34.
7. 辻岡美延・清水和秋　1975　確認的因子分析における検査尺度構成（因子的真実性の原理による項目分析—社会的態度測定における一結果—）関西大学社会学部紀要，**6**(1)，67—82.

（研究担当者　清水　和秋）

項目分析のメインプログラム

```
      SUBROUTINE MOUT, MATINV, ISFT AND PLOTK ARE REQUIRED,
      NR1 AND NR2 ARE THE SIZE OF DIMENSION.
      EVFD IS THE ESTIMATED FACTOR PATTERN MATRIX.
      EVFS IS THE ESTIMATED FACTOR STRUCTURE MATRIX.
      NX IS THE NUMBER OF ITEMS.
      NF IS THE NUMBER OF FACTORS.
      NMI IS NAMES OF ITEMS.
      NMF IS NAMES OF FACTORS.
      DIMENSION EVFP(210,6),EVFS(210,6),X(210),AVX(210),SDX(210),
     D          F(6),AVF(6),SDF(6),CF(6,6),CR(6,6),CW(6,6),ITTL(40),
     I          IZ(210),NMI(210),NMF(6),S(6)
      DATA NR1,NR2/210,6/
      DATA TN,NM,CON,IFIN/20.0,3,0.7,0/
      READ(5,100) NX,NF,EN
  100 FORMAT(2I4,F8.0)
      READ(5,110) ITTL
  110 FORMAT(40A2)
      READ(5,120) (NMF(I),I=I,NF)
  120 FORMAT(20(2X,A2))
      READ(5,120) (NMI(I),I=1,NX)
      READ(5,130) (S(I),I=1,NF)
  130 FORMAT(10F8.0)
      DO 10 I=1,NF
      AVF(I)=0.0
      SDF(I)=0.0
      DO 10 J=1,I
   10 CF(I,J)=0.0
      DO 20 I=1,NX
      AVX(I)=0.0
      SDX(I)=0.0
      DO 20 J=1,NF
   20 EVFS(I,J)=0.0
      NS=0
   01 READ(1) (X(I),I=1,NX)
      IF(X(1).EQ.EN) GO TO 05
      READ(1) (F(I),I=1,NF)
      NS=NS+1
      DO 30 I=1,NF
      F(I)=F(I)/S(I)
      AVF(I)=AVF(I)+F(I)
      SDF(I)=SDF(I)+F(I)**2
      DO 30 J=1,I
   30 CF(I,J)=CF(I,J)+F(I)*F(J)
      DO 40 I=1,NX
      AVX(I)=AVX(I)+X(I)
      SDX(I)=SDX(I)+X(I)**2
      DO 40 J=1,NF
   40 EVFS(I,J)=EVFS(I,J)+X(I)*F(J)
      GO TO 01
   05 FNS=NS
      REWIND 1
      DO 50 I=1,NF
      AVF(I)=AVF(I)/FNS
      SDF(I)=SQRT(SDF(I)/FNS-AVF(I)**2)
      DO 50 J=1,I
      CF(I,J)=(CF(I,J)/FNS-AVF(I)*AVF(J))/(SDF(I)*SDF(J))
   50 CF(J,I)=CF(I,J)
      DO 60 I=1,NX
      AVX(I)=AVX(I)/FNS
      SDX(I)=SQRT(SDX(I)/FNS-AVX(I)**2)
      DO 60 J=1,NF
   60 EVFS(I,J)=(EVFS(I,J)/FNS-AVX(I)*AVF(J))/(SDX(I)*SDF(J))
      WRITE(6,300) ITTL
  300 FORMAT(1H1/10X,40A2)
      WRITE(6,301)
  301 FORMAT(//5X,6HFACTOR,5X,4HMEAN,6X,4HS.D, /)
      DO 62 J=1,NF
   62 WRITE(6,302) J,NMF(J),AVF(J),SDF(J)
  302 FORMAT(5X,I4,1X,A2,1X,F8.3,2X,F8.3)
      WRITE(6,303)
  303 FORMAT(///10X,44HCORRELATIONS BETWEEN ESTIMATED FACTOR SCORES
      CALL MOUT(NR2,NF,NF,CF,NMF,NMF,1,1)
      WRITE(6,300) ITTL
      WRITE(6,304)
  304 FORMAT(//5X,6H ITEM ,5X,4HMEAN,6X,4HS.D, /)
      DO 65 I=1,NX
   65 WRITE(6,302) I,NMI(I),AVX(I),SDX(I)
      WRITE(6,300) ITTL
      WRITE(6,305)
  305 FORMAT(/10X,35HESTIMATED FACTOR STRUCTURE OF ITEMS )
      CALL MOUT(NR1,NX,NF,EVFS,NMI,NMF,1,1)
      DO 70 I=1,NF
      READ(5,150) (CF(I,J),J=1,I)
  150 FORMAT(10F8.0)
      DO 70 J=1,I
   70 CF(J,I)=CF(I,J)
      WRITE(6,300) ITTL
      WRITE(6,306)
  306 FORMAT(/10X,28HCORRELATIONS BETWEEN FACTORS )
      CALL MOUT(NR2,NF,NF,CF,NMF,NMF,1,1)
      DO 75 I=1,NF
      READ(5,150) (CR(I,J),J=1,I)
      DO 75 J=1,I
   75 CR(J,I)=CR(I,J)
      WRITE(6,307)
  307 FORMAT(///10X,37HCORRELATIONS BETWEEN REFERENS VECTORS )
      CALL MOUT(NR2,NF,NF,CR,NMF,NMF,1,1)
      DO 77 I=1,NF
      DO 77 J=1,NF
   77 CW(I,J)=CF(I,J)
      CALL MATINV(NR2,NF,CW,AVF,SDF)
      DO 80 I=1,NX
      DO 80 J=1,NF
      EVFP(I,J)=0.0
      DO 80 K=1,NF
   80 EVFP(I,J)=EVFP(I,J)+EVFS(I,K)*CW(K,J)
      WRITE(6,300) ITTL
      WRITE(6,308)
  308 FORMAT(/10X,33HESTIMATED FACTOR PATTERN OF ITEMS )
      CALL MOUT(NR1,NX,NF,EVFP,NMI,NMF,1,1)
      CALL ISFT(NR1,NX,NF,EN,EVFP,EVFS,AVX,SDX,IZ,NMI,NMF,ITTL,
     S          TN,NM,CON,IFIN,X)
      CALL PLOTK(NR1,NR2,NV,NF,EVFP,-1,0,1,0,0,0,IROW,CR,CF)
      STOP
      END
```

項目選択のサブルーチン

```
      SUBROUTINE ISFT(NR1,NX,NF,EN,EVFP,EVFS,AV,SD,IZ,NMI,NMF,ITTL,
     S                TN,NM,CON,IFIN,X)
      SUBROUTINE ISFT IS FOR THE ITEM SELECTION
                                BY THE PRINCIPLE OF FACTOR TRUENESS.
      EVFD IS THE ESTIMATED FACTOR PATTERN MATRIX.
      EVFS IS THE ESTIMATED FACTOR STRUCTURE MATRIX.
      THE SIZE OF DIMENSION NN,NH AND IH MUST BE LARGER THAN THE MAXIMUM
      NUMBER OF ITEMS WHICH WILL BE SELECTED FOR EACH FACTOR IN ISFT.
      NX IS THE NUMBER OF ITEMS.
      NF IS THE NUMBER OF FACTORS.
      NMI IS NAMES OF ITEMS.
      NMF IS NAMES OF FACTORS.
      DIMENSION EVFP(NR1,1),EVFS(NR1,1),AV(1),SD(1),IZ(1),NMI(1),NMF(1),
     D          NN(15),NH(15),IH(15),JZ(10),AVE(10),X(1),ITTL(40)
      DATA JZ/2H 1,2H 2,2H 3,2H 4,2H 5,2H 6,2H 7,2H 8,2H 9,2H10/
      DATA IHB,IHS,IHB2/1H ,1H*,2H  /
      TN=SIN(TN)/COS(TN)
      DO 1 I=1,NX
    1 IZ(I)=IHB2
      DO 90 K=1,NF
      IF(IFIN.EQ.0) GO TO 2
      READ(5,100) JJ
  100 FORMAT(I4)
      READ(5,101) (NN(I),I=1,JJ)
  101 FORMAT(20I4)
      GO TO 5
    2 JJ=0
      DO 20 I=1,NX
      W=ABS(EVFP(I,K))
      IF(W.LT.CON)  GO TO 20
      KO=0
      DO 10 J=1,NF
      IF(J.EQ.K) GO TO 10
      U=ABS(EVFP(I,J))
      T=U/W
      IF(T.GT.TN)  GO TO 10
      KO=KO+1
   10 CONTINUE
      IF(KO.LT.NM)  GO TO 20
      JJ=JJ+1
      IP=I
      NN(JJ)=IP
      IF(EVFP(I,K).GE.0.0) GO TO 20
      NN(JJ)=-IP
   20 CONTINUE
    5 XNI=JJ
      IF(JJ.EQ.0)  GO TO 86
      DO 30 I=1,NF
   30 AVE(I)=0.0
      DO 40 I=1,JJ
   40 IH(I)=IHB
      DO 70 L=1,NF
      DO 60 I=1,JJ
      J=NN(I)
      NH(I)=NN(I)
      IF(J.GT.0) GO TO 60
      IH(I)=IHS
      J=-J
      NH(I)=J
      IF(L.GT.1)  GO TO 60
      AV(J)=6.0-AV(J)
      DO 50 LL=1,NF
      EVFP(J,LL)=-EVFP(J,LL)
   50 EVFS(J,LL)=-EVFS(J,LL)
   60 AVE(L)=AVE(L)+EVFP(J,L)
   70 AVE(L)=AVE(L)/XNI
      DO 75 I=1,JJ
      KK=I
      DO 74 J=I,JJ
      II=NH(KK)
      JI=NH(J)
      IF(EVFP(II,K)-EVFP(JI,K))  7,74,74
    7 KK=J
   74 CONTINUE
      NZ=NH(I)
      NW=NN(I)
      IW=IH(I)
      NH(I)=NH(KK)
      NN(I)=NN(KK)
      IH(I)=IH(KK)
      NH(KK)=NZ
      NN(KK)=NW
   75 IH(KK)=IW
      WRITE(6,300) ITTL,K
  300 FORMAT(1H1,10X,40A2/5X,6HFACTOR,I4//5X,4HITEM,13X,7HPATTERN  //)
      WRITE(6,301) (NMF(I),I=1,NF)
  301 FORMAT(28X,10(A2,6X))
      OS=0.0
      OT=0.0
      DO 80 I=1,JJ
      J=NN(I)
      IF(J.GT.0)  GO TO 79
      J=-J
   79 CONTINUE
      DO 81 L=1,JJ
      LJ=NH(L)
      DO 81 M=1,NF
   81 OT=OT+SD(J)*EVFS(J,M)*EVFP(LJ,M)*SD(LJ)
      OS=OS+SD(J)*EVFS(J,K)
      WRITE(6,302) IZ(J),I,IH(I),J,NMI(J),(EVFP(J,L),L=1,NF)
  302 FORMAT(1H ,3X,A2,3X,I2,2X,A1,I5,2X,A2,2X,10F8.3)
   80 IZ(J)=JZ(K)
      WRITE(6,303) (AVE(L),L=1,NF)
  303 FORMAT(//25X,10F8.3)
      RFT=OS/SQRT(OT)
      WRITE(6,305) RFT
  305 FORMAT(////25X,30HCOEFFICIENT OF FACTOR-TRUENESS  /
     F        35X,5HRFT =,F10.4)
      CALL SPC(NR1,JJ,NX,NF,EN,EVFP,EVFS,AV,SD,NN,NH,K,X,ITTL)
      DO 85 I=1,JJ
      J=NN(I)
      IF(J.GT.0)  GO TO 85
      J=-J
      AV(J)=6.0-AV(J)
      DO 84 LL=1,NF
      EVFP(J,LL)=-EVFP(J,LL)
   84 EVFS(J,LL)=-EVFS(J,LL)
   85 CONTINUE
      GO TO 90
   86 WRITE(6,300) ITTL,K
      WRITE(6,304)
  304 FORMAT(1H0//10X,50HNO ITEMS HAVE SUFFICIENT LOADINGS ON THIS FACTO
     *R.   )
   90 CONTINUE
      RETURN
      END
```

構成尺度統計量のサブルーチン

```
      SUBROUTINE SPC(NR1,JJ,NX,NF,EN,EVFP,EVFS,AV,SD,NN,NH,K,X,ITTL)
      SUBROUTINE SPC IS FOR THE FACTOR STRUCTURE, THE FACTOR PATTERN AND
                              THE COMMUNARITY OF CONSTRUCTED SCALE,
      R IS THE CORRELATIONS BETWEEN ITEMS,
      EVFD IS THE ESTIMATED FACTOR PATTERN MATRIX,
      EVFS IS THE ESTIMATED FACTOR STRUCTURE MATRIX,
      NX IS THE NUMBER OF ITEMS.
      NF IS THE NUMBER OF FACTORS.
      DIMENSION EVFP(NR1,1),EVFS(NR1,1),NH(1),NN(1),AV(1),SD(1),
     D          R(15,15),F(6),XX(15),X(1),ITTL(40)
      DATA NR3/15/
      REWIND 1
      DO 10 I=1,JJ
      DO 10 J=1,I
   10 R(I,J)=0.0
      NS=0
    1 READ(1) (X(I),I=1,NX)
      IF(X(1).EQ.EN)  GO TO 3
      NS=NS+1
      READ(1) (F(I),I=1,NF)
      DO 20 I=1,JJ
      L=NN(I)
      IF(L.GT.0) GO TO 2
      L=-L
      XX(I)=6.0-X(L)
      GO TO 20
    2 XX(I)=X(L)
   20 CONTINUE
      DO 30 I=1,JJ
      DO 30 J=1,I
   30 R(I,J)=R(I,J)+XX(I)*XX(J)
      GO TO 1
    3 FNS=NS
      REWIND 1
      DO 40 I=1,JJ
      LI=NH(I)
      DO 40 J=1,I
      LJ=NH(J)
      R(I,J)=(R(I,J)/FNS-AV(LI)*AV(LJ))/(SD(LI)*SD(LJ))
   40 R(J,I)=R(I,J)
      OS=0.0
      OP=0.0
      VARI=0.0
      OT=0.0
      DO 50 I=1,JJ
      L=NH(I)
      OS=OS+SD(L)*EVFS(L,K)
      OP=OP+SD(L)*EVFP(L,K)
      DO 50 J=1,JJ
      M=NH(J)
      VARI=VARI+SD(L)*R(I,J)*SD(M)
      DO 50 KK=1,NF
   50 OT=OT+SD(L)*EVFS(L,KK)*EVFP(M,KK)*SD(M)
      OS=OS/SQRT(VARI)
      OP=OP/SQRT(VARI)
      OT2=OT/VARI
      OT=SQRT(OT2)
      RFT=OS/OT
      WRITE(6,300) ITTL
  300 FORMAT(1H1/10X,40A2)
      WRITE(6,301) OS
  301 FORMAT(1H0//5X,34HCOEFFICIENT OF FACTORIAL VALIDITY
     F        18H(FACTOR STRUCTURE) ,/10X,4HOS =,F10.4)
      WRITE(6,302) OP
  302 FORMAT(1H0//5X,14HFACTOR PATTERN    /10X,4HOP =,F10.4)
      WRITE(6,303) OT2,OT
  303 FORMAT(1H0//5X,25HCOMMUNALITY OF THIS SCALE  /
     F        10X,4HH2 =,F10.4,10X,7HOT(H) =,F10.4)
      WRITE(6,304) RFT
  304 FORMAT(1H0//5X,31HCOEFFICIENT OF FACTOR-TRUENESS    /
     F        10X,5HRFT =,F10.4)
      WRITE(6,300) ITTL
      WRITE(6,305)
  305 FORMAT(1H0//9X,4HMEAN,4X,4HS,D,  /)
      DO 60 I=1,JJ
      J=NH(I)
   60 WRITE(6,306) I,AV(J),SD(J),J
  306 FORMAT(3X,I2,2F8.3,3X,I3)
      WRITE(6,307)
  307 FORMAT(1H0//10X,40HCORRELATIONS BETWEEN ITEMS ON THIS SCALE    )
      CALL MOUT(NR3,JJ,JJ,R,0,0,0,0)
      RETURN
      END
```